SAMPLE EXERCISES USING MAPLE

N. P. STRICKLAND

Polynomials

Exercise 1. Expand out the following products:

$$(x-1)(1+x)$$
$$(x-1)(1+x+x^2)$$
$$(x-1)(1+x+x^2+x^3)$$
$$(x-1)(1+x+x^2+x^3+x^4)$$

What would be the next two equations in this sequence? What is the general rule? (Try to write down your answer as a complete, self-contained statement that could be understood by someone who had not seen the calculation before.)

Exercise 2. Consider the following polynomials

$$\begin{aligned}
\phi_1(x) &= x-1 \\
\phi_2(x) &= x+1 \\
\phi_3(x) &= x^2+x+1 \\
\phi_4(x) &= x^2+1 \\
\phi_6(x) &= x^2-x+1 \\
\phi_{12}(x) &= x^4-x^2+1.
\end{aligned}$$

Expand out the following products:

$$\phi_1(x)\phi_2(x)$$
$$\phi_1(x)\phi_3(x)$$
$$\phi_1(x)\phi_2(x)\phi_4(x)$$
$$\phi_1(x)\phi_2(x)\phi_3(x)\phi_6(x).$$

Can you see the pattern? Can you guess what is the corresponding equation involving $\phi_{12}(x)$? Check your guess using Maple.

Remark 3. These functions are called *cyclotomic polynomials*; they are important in Number Theory, and have many interesting properties. There are various different ways to define them, one of which is based on the pattern that you should have observed.

Exercise 4. Factor the polynomials x^2-1, x^4-1, x^8-1 and $x^{16}-1$ using Maple's factor() command. What is the pattern? What is the next thing in the sequence? Check your answer with Maple.

Date: November 21, 2003.

Exercise 5. Put $u = a + b$ and $v = ab$. Expand out the following expressions:

$$u^2 - 2v$$
$$u^3 - 3uv$$
$$u^4 - 4u^2v + 2v^2$$
$$u^5 - 5u^3v + 5uv^2$$
$$u^6 - 6u^4v + 9u^2v^2 - 2v^3.$$

You should do the first three by hand, and get Maple to do the remaining two. You should see an obvious pattern in your answers. (There is a pattern in the questions as well, but it is a lot harder to describe.)

Graphs of powers

Exercise 6. Plot the function $y = x^3$ for $-1 \leq x \leq 1$, like this:

```
> plot(x^3,x=-1..1);
```

Now plot the functions $y = x, x^2, ..., x^6$ together, like this:

```
> plot({x,x^2,x^3,x^4,x^5,x^6},x=-1..1);
```

Describe the pattern in words. What do you think that the graph of x^{100} looks like? How about x^{101}? Check your answer like this:

```
> plot({x,x^2,x^3,x^4,x^5,x^6,x^100,x^101},x=-1..1);
```

Exercise 7. Plot the functions x^5 and $x^{1/5}$ together. In this case, it is convenient to ensure that the graph is square, with the same units on the two axes; for this we use the option "scaling=constrained".

```
> plot({x^5,x^(1/5)},x=0..1,scaling=constrained);
```

What is the relationship between the two graphs? Try it again with the 5 replaced by 6. Then just plot the graph of x^7, and draw by hand the graph of $x^{1/7}$ (on paper or in your imagination).

Decaying oscillations

Consider the function $y = \exp(-t)\sin(20t)$. We first plot it to get a feel for what it looks like:

```
> plot(exp(-t) * sin(20*t),t=0..10);
```

The function oscillates rapidly, but also gets smaller and smaller, becoming very small when t gets large. This is the sort of response that you get if you pluck a guitar string; it vibrates, but the vibrations die away. You get similar behaviour with water waves, oscillations in electrical circuits, and almost any other physical system that you can think of. Of course, the exact formula will be a bit different in each case, depending on the size and frequency of the oscillations and the rate of decay. The most general type of function that we will consider is

$$y = a\exp(-bt)\sin(ct),$$

where a, b and c are constants. What difference do these constants make? To investigate, enter the following in Maple.

```
> a := 2; b := 1.5; c := 20;
> plot(a * exp(-b*t) * sin(c*t),t=0..5,y = -2..2);
```

Now go back to the first line, change some of the numbers, press RETURN, then go to the "plot" line and press RETURN again to plot the new function. First change a several times, then change b several times, then c.

You should conclude that a just affects the size of the oscillations (which is easy to see anyway). Increasing b makes the oscillations decay more quickly, and increasing c makes the oscillations more rapid.

From now on, we want to treat a, b and c as symbols, not as particular numbers:

```
> unassign('a','b','c');
```

We next calculate the derivatives of y. We can write y in the form $y = uv$, where $u = a\exp(-bt)$ and $v = \sin(ct)$. You should remember that

$$\frac{du}{dt} = \frac{d}{dt} a\exp(-bt) = -ab\exp(-bt)$$
$$\frac{dv}{dt} = \frac{d}{dt}\sin(ct) = c\cos(ct).$$

We also have the rule for differentiating a product of two things:

$$\frac{d}{dt}(uv) = u\frac{dv}{dt} + v\frac{du}{dt}.$$

This gives

$$y' = \frac{dy}{dt} = ac\exp(-bt)\cos(ct) - ab\exp(-bt)\sin(ct).$$

You can check this in Maple as follows:

```
> y := a*exp(-b*t)*sin(c*t);
> diff(y,t);
```

(As quite often happens, Maple gives the terms in a slightly unnatural order, but nonetheless you can check that tey match up with what we wrote above.)

Next, you can compute the second derivative of y:

```
diff(y,t,t);
```

The answer contains three terms, all a bit like the ones we have seen before. There is in fact a precise relationship between y, y' and y'', given by the equation

$$y'' + 2by' + (b^2 + c^2)y = 0.$$

You can check this in Maple: `expand(diff(y,t,t) + 2*b*diff(y,t) + (b^2+c^2) * y);` This identity is a *differential equation* for y. Physically speaking, our whole discussion is really the wrong way around. The physical theory gives the differential equation that we ended with, and one has to solve the equation to find the formula that we started with.

Gaussians

In this example, we consider variants of the function $f(x) = \exp(-x^2)$. We first get the general picture:

```
> plot(exp(-x^2),x=-5..5);
```

We have a hump of height one, centred at $x = 0$. The function is nicely symmetrical. It seems to be zero when x is bigger than 3 or so, and is never negative. The curve is reasonably close to a triangle with vertices at $x = \pm 2$ on the x-axis, and at $y = 1$ on the y-axis. The area of a triangle is half the base length times the height, which in our case is $\frac{1}{2} \times 4 \times 1 = 2$, so the area under the curve is roughly 2. At $x = 0$ the curve is horizontal, meaning that the derivative $f'(0)$ is zero. The curve is flat when x is large, so again it looks like $f'(x) = 0$ when x is large.

It is important to be able to make observations like these from the graph, but it is equally important to be able to prove them mathematically using formulae.

(1) Where is the hump, and how big is it? Clearly $f(0) = \exp(0) = e^0 = 1$. For any $x \neq 0$ we have $-x^2 < 0$ and so $\exp(-x^2) = 1/e^{x^2} < 1$. Thus, the maximum value is 1, and this value occurs when $x = 0$. Another approach is to say that the maximum is found where the derivative is zero; we will discuss the derivative later.

(2) To say that the function is symmetrical around the y-axis just means that $f(-x) = f(x)$. This is easy to verify, because $(-x)^2 = x^2$ and so $-(-x)^2 = -x^2$ and so $f(-x) = \exp(-(-x)^2) = \exp(-x^2) = f(x)$.

(3) Is it really true that $f(x) = 0$ when $x \geq 3$? We have $f(3) = e^{-9} = 1/e^9$. The number e^9 is large (about 8000) but not infinite, so $1/e^9$ cannot be zero. Take another look at the graph, concentrating on the region where $3 \leq x \leq 4$:

```
> plot(exp(-x^2),x=3..4);
```

The values are small, but strictly positive.

(4) Is it true that $f(x)$ is never negative? As $e > 0$ we have $e^{x^2} > 0$ for all x and so $f(x) = 1/e^{x^2} > 0$ as claimed.

(5) What is the area under the curve? Recall that the area under a curve is just the definition (or at least one of the possible definitions) of integration. The area that we want is given by:

$$\int_{-\infty}^{\infty} f(x)dx$$

What is the integral of $f(x)$? Is it perhaps something like $\exp(-x^2)/(-2x)$? If you think you know an integral but you are not sure, there is an easy way to check: just differentiate, and see if you get back to the function you started with. We can do this by hand, or ask Maple:

```
> diff(exp(-x^2)/(-2*x),x);
```

The answer is not equal to $f(x)$, so our integration was incorrect. We can just ask Maple to do the integration instead:

```
> int(exp(-x^2),x);
```

The answer involves something called $\text{erf}(x)$, which you probably have not seen before. This is not so surprising; Maple knows a long list of strange functions, many of them unknown even to most professional mathematicians (although erf itself is only moderately obscure.) If you are curious, try

```
> help(erf);
```

In any case, we do not need the general value of the integral, just the integral from $-\infty$ to ∞, and it turns out that this can be given in familiar terms:

```
> int(exp(-x^2),x=-infinity..infinity);
```

The answer is really quite amazing: where on earth did the π come from? Let us check the numerical value:

```
> evalf(sqrt(Pi));
```

The answer is about 1.77; recall that our crude estimate with a triangle was 2, which is not great, but not too bad either.

(6) We next consider the derivative of our function $y = f(x)$. This can be calculated with the chain rule: we put $v = -x^2$, so $y = \exp(v)$ and so

$$\begin{aligned} \frac{dy}{dv} &= \exp(v) \\ \frac{dv}{dx} &= -2x \\ f'(x) = \frac{dy}{dx} &= \frac{dy}{dv}\frac{dv}{dx} = -2x\exp(v) \\ &= -2x\exp(-x^2). \end{aligned}$$

As $\exp(-x^2)$ is never zero, we see that $f'(x)$ is only zero when $x = 0$. This is another way to see that the maximum is at $x = 0$.

You may be aware that the function $f(x)$ (or related functions such as $a\exp(-(x-b)^2/c)$) are very important in statistics, as the probability distributions for random variables. It is less well-known that these functions also occur in the theory of temperature: if you heat a long bar in the middle and then let the heat spread out for a while, then the temperature profile will be given by a function like $f(x)$. Of course, the profile will change over time: as the heat spreads out, the hump will get lower and broader. If we take the time dependence into account, the temperature is actually given by a function like

$$u(x,t) = \exp(-x^2/4t)/\sqrt{t}.$$

Let us plot this function, to check that the formula seems reasonable:

```
> u := ((x,t) -> exp(-x^2/(4*t))/sqrt(t));
> plots[animate](u(x,t),x=-50..50,t=0.5..100);
```

You will see a picture of the hump as it is at time $t = 0.5$. If you click on the plot, a box will appear around it and some new controls will appear at the top of your Maple window, like the controls on a CD player. Click on the "play" button to see how the plot changes over time; it spreads downwards and outwards, as expected.

Just as in the case of damped oscillations, the formula for u really comes from solving a differential equation. As this is a course in mathematics rather than physics, we allow ourselves to work backwards and just check the differential equation as a mathematical exercise.

We can differentiate u just as before, by putting $v = -x^2/4t$ (so $u = \exp(v)/\sqrt{t}$) and using the chain rule. We have

$$\frac{dv}{dx} = \frac{d}{dx}\left(\frac{-x^2}{4t}\right) = \frac{-2x}{4t} = \frac{-x}{2t}.$$

$$\frac{du}{dx} = \frac{du}{dv}\frac{dv}{dx} = \frac{\exp(v)}{\sqrt{t}}\frac{-x}{2t} = \frac{-x\exp(-x^2/4t)}{2t^{3/2}}.$$

Here we have just carried the t through treating it as though it were a constant. This just means that we are thinking about how u varies along the bar at a particular moment in time, and not worrying about how u changes with t. This is called a *partial derivative*, because we are only measuring part of the variation of u. It is usual to change notation slightly to indicate this, and write $\partial u/\partial x$ instead of du/dx to indicate this. Thus, our conclusion is that

$$\frac{\partial u}{\partial x} = \frac{-x\exp(-x^2/4t)}{2t^{3/2}}.$$

We can instead consider a fixed point on the bar, and consider how the temperature at that point varies over time. This means that we treat x as a constant, and differentiate with respect to t:

```
> diff(u,t);
```

I claim that u satisfies the differential equation

$$\frac{\partial u}{\partial t} - \frac{\partial^2 u}{\partial t^2} = 0.$$

This is easy to check with Maple:

```
> simplify(diff(u,t) - diff(u,x,x));
```

Can you verify this by hand?

Mobius functions

A Möbius transformation is a function of the form

$$f(x) = \frac{ax+b}{cx+d}$$

where a, b, c and d are nonzero numbers with $ad > bc$. (Here we have slightly changed the proper definition, to make our lives easier.)

A typical example is the function $f(x) = (x-1)/(x+1)$ (here $a = 1$, $b = -1$, $c = 1$ and $d = 1$). Let us look at the graph:

```
> f := (x) -> (x-1)/(x+1);
> plot(f(x),x=-10..10,y=-10..10);
```

Something funny is happening at $x = -1$. That seems reasonable, because the formula gives $f(-1) = -2/0$, which is undefined. You might want to say that $-2/0 = \infty$, but all statements about infinity are fraught with difficulty and subtle pitfalls. It is hard even to say things that are truly meaningful, and harder still to say things that are actually correct. The issues involved are in fact well understood (this is one of the great triumphs of 19th century mathematics) but we will leave them for future courses (starting with PMA113, "Introduction to Analysis"). For the moment, we will simply say that $f(-1)$ is undefined.

When x is very large, the curve appears to flatten off, but it is not too easy to see the level that it is tending to. To work this out, we rearrange the formula a little:

$$f(x) = \frac{x-1}{x+1} = \frac{(x+1)-2}{x+1} = 1 - \frac{2}{x+1}.$$

When x is large (either positive or negative) then $2/(x+1)$ will be very small, so $f(x)$ will be close to 1. We can try this numerically, either directly:

```
> f(1000.0);
```

or by plotting the graph for large values of x:

```
> plot(f(x),x=1000..2000);
```

Alternatively, we can ask Maple to work out the limit symbolically:

```
> limit(f(x),x=infinity);
```

It is also interesting to work out the derivative of $f(x)$. We can write $f(x) = u(x)/v(x)$, where $u(x) = x - 1$ and $v(x) = x + 1$. The derivative is then given by the quotient rule:

$$f'(x) = \frac{u'(x)v(x) - v'(x)u(x)}{v(x)^2} = \frac{1.(x+1) - 1.(x-1)}{(x+1)^2} = \frac{2}{(x+1)^2}$$

If you ask Maple to do this, you will get a more complicated expression, and you will need to simplify it explicitly to get our answer:

```
> diff(f(x),x);
> simplify(%);
```

Here is a puzzle. As $(x+1)^2$ is never negative, the derivative as always positive, so the function slopes upwards and is always increasing. Out at the left hand end of the graph, x is a large, negative number, so $f(x)$ is approximately 1. The function increases from there, so $f(x)$ should always be at least 1. But $f(0) = (0-1)/(0+1) = -1$. What is going on?

The answer is clear from the pictures that we drew earlier. The graph does indeed slope upwards everywhere, but it also has a massive jump at $x = -1$. The idea that a function with positive derivative is increasing relies on there being no jumps. You will study such questions in more detail in PMA113, but for the moment, you should just keep an eye open for the kinds of things that can happen when infinities crop up.

Now suppose we have a Möbius transformation $g(x) = (ax+b)/(cx+d)$, but we do not know what a, b, c and d are. Then $g(0) = b/d$ and we do not know what b/d is, so we cannot even begin to plot the graph. Instead, we must work with formulae to understand the features of the function.

(1) When is $g(x)$ undefined? A problem occurs if we try to divide by zero, which happens when $cx+d=0$, or equivalently $x = -d/c$. The formula $g(x) = (ax+b)/(cx+d)$ is fine for all other values of x.

(2) What happens when x is very large? You will see more about this in PMA113, but this example is quite easy to understand directly. The trick is to rewrite the formula slightly, by dividing the top and bottom of the fraction by x:

$$g(x) = \frac{ax+b}{cx+d} = \frac{a+b/x}{c+d/x}.$$

When x is very large, b/x and d/x will be very small so we can neglect them and we see that $g(x) \simeq a/c$.

(3) What is the derivative? We can again use the quotient rule:

$$g'(x) = \frac{a(cx+d) - c(ax+b)}{(cx+d)^2} = \frac{acx + ad - acx - bd}{(cx+d)^2} = \frac{ad-bc}{(cx+d)^2}.$$

We assumed at the beginning of this section that $ad > bc$, so $ad-bc$ is positive, and $(cx+d)^2$ is nonnegative (because it is the square of something). We have the usual problem when $x = -d/c$ because then $cx+d = 0$ and our formula $g'(x) = (ad-bc)/(cx+d)^2$ involves division by zero and so is not defined. For all other values of x, we see that $g'(x) > 0$ and so $g(x)$ is increasing. We conclude that the picture is the same as for the special case that we studied before. The function starts off at approximately a/c when x is very large and negative, then increases to ∞ as we approach $x = -d/c$, jumps instantly down to $-\infty$ and then increases back to a/c as x becomes large and positive.

Star Trek

In this section we will measure speed in terms of warp factors, as in Star Trek. Warp factor 1 is the speed of light; 70 mph is warp factor 0.0000001. Sadly for Star Trek, warp factors greater than one are forbidden by Einstein's theory of relativity, and do not occur in the real world.

Remark 8. Warp factor 1 is about a foot per nanosecond. A nanosecond is a billionth of a second, which is the clock cycle time of a computer chip running at 1 GHz, which is fairly typical in 2001.

Suppose we have an object moving at warp factor v. If v is close to 1 (ie the object is approaching the speed of light), then all sorts of strange relativistic effects occur: time slows down, lengths contract, light changes colour, and so on. The strength of these effects is given by the function

$$g(v) = (1-v^2)^{-1/2} = \frac{1}{\sqrt{1-v^2}}.$$

In this section, we investigate this function as a purely mathematical exercise.

First, we draw the graph:

```
> g := (v) -> (1-v^2)^(-1/2);
> plot(g(v),v=-2..2,y=0..10);
```

(1) First, note that the graph is symmetrical, ie $g(v) = g(-v)$. This is obvious from the formula, because $(-v)^2 = v^2$, and is also physically reasonable: the strength of relativistic effects does not depend on the direction that you are moving in.
(2) Next, we have $g(0) = 1$. This means that if you are not moving (ie $v = 0$) then time slows down by a factor of 1 (ie it does not slow down at all). This seems very reasonable.
(3) As v approaches 1 we see that $1 - v^2$ approaches 0, so $\sqrt{1 - v^2}$ also approaches 0, so $g(v)$ tends to infinity, and the relativistic effects become extremely strong.
(4) What has happened to the graph for $v > 1$? If $v > 1$ then $v^2 > 1$ so $1 - v^2 < 0$. We cannot take the square root of a negative number, so $g(v)$ is undefined. This is reasonable, given Einstein's dictum that speeds greater than warp 1 are impossible. (If you know about complex numbers, you may want to say that $g(v)$ is defined but imaginary. This is mathematically sensible but not physically meaningful.)
(5) We next look at $g(v)$ when v is small. Of course we then have $g(v) \simeq g(0) = 1$, but we want something a bit more accurate than that. We will use the approximation given by Taylor's theorem:

$$g(v) \simeq g(0) + g'(0)v + g''(0)v^2/2$$

We've seen that $g(0) = 0$. The function has a minimum at $v = 0$, so we must have $g'(0) = 0$, but we will check this algebraically anyway.

We first work out the relevant derivatives:

```
> diff(g(v),v);
> diff(g(v),v,v);
```

We then work out their values at $v = 0$.

```
> eval(diff(g(v),v),v=0);
> eval(diff(g(v),v,v),v=0);
```

Putting this back into Taylor's formula gives

$$g(v) \simeq 1 + v^2/2.$$

We could actually have asked Maple this more directly:

```
> series(g(v),v,3);
```

The motorway speed limit is approximately $v = 10^{-7}$, so $g(v) \simeq 1 + 5 \times 10^{-15}$. Your car clock is slowed by a factor of $g(v)$, which works out to about 0.15 microseconds per year. This kind of effect is measurable with atomic clocks.

Absolute zero

There is a certain temperature called "absolute zero" (about -273 degrees centigrade) that is the coldest that anything can possibly get. (Warmth is caused by random movement of molecules; at absolute zero they are not moving at all, so it is not possible to cool down any further.) We'll use a scale of temperature in which $t = 0$ corresponds to absolute zero. In this context, the following function comes up regularly:

$$f(t) = \exp(-1/t) = 1/\exp(1/t).$$

It has some interesting mathematical properties, which we will now investigate. We start by plotting the graph:

```
> f := (t) -> exp(-1/t);
> plot(f(t),t=0..1);
```

(We have only drawn the curve for $t > 0$, because negative temperatures do not make sense.)

The main feature of interest is that the curve is very flat near $t = 0$, which is a reflection of the rather strange nature of absolute zero. To see this more clearly, plot the graph for $0 \leq t \leq 0.1$.

How can we understand this mathematically?

(1) Firstly, what is $f(0)$? We cannot just put $t = 0$ in the formula, because then we would have a term $1/0$ which is undefined. Instead we need to look at what happens as t approaches 0. For this, we need to recall the formula for $\exp(x)$:

$$\exp(x) = 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \cdots ,$$

so

$$\exp(1/t) = 1 + t^{-1} + t^{-2}/2 + t^{-3}/6 + \cdots .$$

As t is positive, all the terms are positive, so we have

$$\exp(1/t) = t^{-1} + \text{other positive terms} \geq t^{-1}.$$

It follows that

$$0 \leq f(t) = 1/\exp(1/t) \leq t.$$

If you like, you can check this by plotting $f(t)$ and t together:

```
> plot({f(t),t},t=0..1);
```

In any case, as $0 \leq f(t) \leq t$, it is clear that $f(t)$ goes to zero as t goes to zero.

(2) Next, the curve seems to be flat at $x = 0$, which means that $f'(0)$ should be 0. We can differentiate f by the chain rule:

$$f'(t) = \exp(-1/t)\frac{d}{dt}\left(\frac{-1}{t}\right) = t^{-2}\exp(-1/t) = t^{-2}f(t).$$

What happens as t approaches zero? Certainly t^{-2} tends rapidly to infinity, but it is multiplied by $f(t)$, which (as we saw above) tends to 0. Naively we have something like $\infty \times 0$, which is horribly undefined. We need to be more careful.

The solution is to reuse the method of the previous section. From the formula for $\exp(1/t)$, we see that

$$\exp(1/t) = \frac{t^{-3}}{6} + \text{other positive terms} \geq \frac{t^{-3}}{6} = \frac{1}{6t^3}$$

It follows that

$$0 \leq f(t) = 1/\exp(1/t) \leq 6t^3,$$

and so

$$0 \leq t^{-2}f(t) \leq 6t.$$

If you like, you can check this by plotting $t^{-2}f(t)$ and $6t$ together:

```
> plot({f(t)/t^2,6*t},t=0..1);
```

In any case, as $0 \leq f'(t) = t^{-2}f(t) \leq 6t$, it is clear that $f'(t)$ goes to zero as t goes to zero, as expected.

(3) In the last section, we had a "competition" between $f(t)$ and t^{-2}. The factor $f(t)$ was tending to zero, and t^{-2} was tending to infinity, and $f(t)$ "won" because the product $t^{-2}f(t)$ turned out to tend to zero. What would happen if we replaced t^{-2} by something that tended to infinity much faster, like t^{-5} or t^{-10}? We can investigate by plotting the graphs of $t^{-k}f(t)$ for various values of k:

```
> plot(f(t)/t^4,t=0..1);
> plot(f(t)/t^6,t=0..1);
> plot(f(t)/t^10,t=0..1);
> plot(f(t)/t^20,t=0..1);
```

In each case, the graph becomes very large at around $t = 1/k$, but it then gets small again as t decreases, and tends to 0 at $t = 0$: the $f(t)$ factor definitely wins. We can verify this mathematically by the same method as above: in the full series for $\exp(1/t)$, one of the terms is $t^{-k-1}/(k+1)!$, so

$$\exp(1/t) \geq \frac{t^{-1-k}}{(k+1)!} = \frac{1}{(k+1)!t^{k+1}}.$$

It follows that

$$0 \leq f(t) = 1/\exp(1/t) \leq (k+1)!t^{k+1}$$

so

$$0 \leq t^{-k}f(t) \leq (k+1)!t$$

so $t^{-k}f(t)$ tends to 0 as t tends to 0.

This is the precise sense in which the graph is "very flat" at $t = 0$; it tends to zero more rapidly than t^k for any k.

Remark 9. The statement that "all movement stops at absolute zero" ignores some important complications from quantum mechanics; but we will not go into details here.

Matrices

Exercise 10. (a) Consider the following matrix (called a Vandermonde matrix):

$$A = \begin{pmatrix} 1 & 1 & 1 \\ a & b & c \\ a^2 & b^2 & c^2 \end{pmatrix}$$

Write down the transpose matrix A^T, then calculate AA^T by hand. Can you see the pattern? You may want to introduce some new notation, so that you do not have to write the same expression several times over.

(b) Now consider the matrix

$$B = \begin{pmatrix} 1 & 1 & 1 & 1 \\ a & b & c & d \\ a^2 & b^2 & c^2 & d^2 \\ a^3 & b^3 & c^3 & d^3 \end{pmatrix}$$

Can you guess what BB^T looks like? Check your guess with Maple.

(c) Use Maple to calculate the determinant of B, and factorize it.

(d) What is the determinant of B when $a = b$? Give an answer based on just looking at the matrix, and another answer based on part (c).

Exercise 11. Consider the following matrix:

$$A = \begin{pmatrix} a & b & c \\ b & c & a \\ c & a & b \end{pmatrix}$$

(a) Calculate $\det(A)$ by hand.

(b) Ask Maple to factor $\det(A)$.

(c) What is $\det(A)$ when $a+b+c = 0$? Give an answer based on part (b), and another answer based on row operations.

Exercise 12. Consider the following matrix:

$$A = \begin{pmatrix} 0 & a & b \\ -a & 0 & c \\ -b & -c & 0 \end{pmatrix}$$

(a) What is $A + A^T$?

(b) What is $\det(A)$?

(c) The answer to (b) actually follows from the answer to (a) – can you see why? (This is quite challenging)

Exercise 13. Consider the following matrix (called a Jordan block):

$$A = \begin{pmatrix} -1/a & 1 & 0 & 0 \\ 0 & -1/a & 1 & 0 \\ 0 & 0 & -1/a & 1 \\ 0 & 0 & 0 & -1/a \end{pmatrix}$$

Ask Maple to find A^{-1}, and then check by hand that $AA^{-1} = I$ and $A^{-1}A = I$. Calculate $\det(A)$ by hand.

Exercise 14. Consider the following matrix:

$$A = \begin{pmatrix} 1 & 0 & 1 & 0 \\ a & 1 & c & 1 \\ b & a & d & c \\ 0 & b & 0 & d \end{pmatrix}$$

Check that

$$\det(A) = (b-d)^2 + (a-c)(ad-bc).$$

Exercise 15. Find the inverse of the following matrix:

$$A = \begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 0 \\ 1 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

Exercise 16. Consider the following matrix:

$$A = \begin{pmatrix} 1 & 2 & 1 & 0 & 0 & 0 \\ 1 & 3 & 3 & 1 & 0 & 0 \\ 1 & 4 & 6 & 4 & 1 & 0 \\ 1 & 5 & 10 & 10 & 5 & 1 \end{pmatrix}$$

Do you recognize the entries? Row-reduce the matrix.

Exercise 17. Suppose that $x^2 + y^2 + z^2 = 1$ (make sure that you use this fact to simplify all your answers). Consider the matrix

$$A = \begin{pmatrix} x^2 & xy & xz \\ xy & y^2 & yz \\ xz & yz & z^2 \end{pmatrix}$$

(a) What is $\det(A)$?
(b) What is $\operatorname{trace}(A)$?
(c) What is A^T?
(d) What is A^2?

Quaternions

Define

$$Q(a,b,c,d) = \begin{pmatrix} a & -b & -c & -d \\ b & a & -d & c \\ c & d & a & -b \\ d & -c & b & a \end{pmatrix}$$

(Matrices like this are called *quaternion matrices.*) We can teach the rule to Maple as follows:

```
> Q := (a,b,c,d) ->
    matrix([[a,-b,-c,-d],[b,a,-d,c],[c,d,a,-b],[d,-c,b,a]]);
```

Get Maple to calculate the product

$$P = Q(a,b,c,d)Q(w,x,y,z).$$

Let p, q, r and s be the entries in the first column of P. If you check carefully, you will see that the entry at the top of the second column is the same as $-q$. Similarly, every other entry in P is either $\pm p$, $\pm q$, $\pm r$ or $\pm s$, so the whole matrix can be written in terms of p, q, r and s. Do this, and thus check that $P = Q(p,q,r,s)$.

Euler angles

Consider the matrix

$$A = \begin{pmatrix} \cos(\theta)\cos(\phi) & -\sin(\theta) & -\cos(\theta)\sin(\phi) \\ \sin(\theta)\cos(\phi) & \cos(\theta) & -\sin(\theta)\sin(\phi) \\ \sin(\phi) & 0 & \cos(\phi) \end{pmatrix}$$

(a) Calculate $\det(A)$, using the identities

$$\sin(\theta)^2 + \cos(\theta)^2 = \sin(\phi)^2 + \cos(\phi)^2 = 1$$

to simplify your answer.

(b) Calculate the inverse of A by the cofactor method. Check that $A^{-1} = A^T$.

Quadratics

Let a, b and c be real numbers, and put

$$f(t) = t^2 - (a+c)t + (ac - b^2).$$

This is a quadratic function, so it might have no real roots, or one real root, or two real roots, and if it has any real roots, they might be positive or negative or zero.

For functions $f(t)$ as above, it turns out that not all of these possibilities can occur.

(a) Enter the following in Maple:

```
> f := (t) -> t^2 - (a + c) * t + (a * c - b^2);
```

Now, if you want to see the graph of $f(t)$ when $a = 1$, $b = 2$ and $c = -3$ you can do

```
> a := 1; b := 2; c := -3;
> plot(f(t),t=-10..10);
```

You will see that there are two real roots, one negative and one positive. Try some other values of a, b and c, and see what happens. (You may also want to plot a different range of values instead of $-10 \leq t \leq 10$.)

(b) Find the number t_0 where $f'(t_0) = 0$. Where is t_0 in the above pictures?

(c) Show that $f(t_0) = -(a-c)^2/4 - b^2$. When is this equal to zero?

(d) Show that if $a = c$ and $b = 0$, then $f(t)$ has precisely one real root, but for all other values of a, b and c, the function $f(t)$ has two different real roots.

(e) Suppose that $a > 0$ and $ac - b^2 > 0$. Show that $c > 0$ and thus that $t_0 > 0$. Observe that $f(0) = ac - b^2 > 0$. By looking through the pictures for cases that are compatible with these facts, show that all the roots of $f(t)$ are real and positive.

(f) Conversely, suppose that all the roots of $f(t)$ are real and positive. This means that there are numbers $\lambda, \mu > 0$ such that $f(t) = (t-\lambda)(t-\mu)$. Show that $ac - b^2 > 0$ and $a + c > 0$. Deduce that $ac > 0$, and say what this means about the signs of a and c. Given what you know about the sign of $a + c$, deduce that $a > 0$. This gives the converse of part (e).

Remark 18. This exercise is really about the matrix $A = \left(\begin{smallmatrix} a & b \\ b & c \end{smallmatrix}\right)$. Note that A is symmetric, ie $A^T = A$. Note also that a is the top left entry in A, and that $ac - b^2$ is the determinant, and that $f(t)$ is the characteristic polynomial. Thus parts (e) and (f) say that the eigenvalues of A are positive iff the determinant and the top left entry are positive. This can be generalized as follows. Let A be a real symmetric $n \times n$ matrix; then all eigenvalues of A are real. Now let A_k be the submatrix formed by the first k rows of the first k columns, so A_k is a real symmetric $k \times k$ matrix. Then it turns out that all eigenvalues of A are positive iff $\det(A_k) > 0$ for all k. It would be very painful to prove this directly, as we did in the case $n = 2$. Instead, we use more abstract methods, covered in the Level 2 Linear Mathematics courses.

Sequences

Exercise 19. Fix two positive real numbers a and b. Put

$$x_n = (a^{n+1} + b^{n+1})/(a^n + b^n)$$
$$y_n = (a^n b + ab^n)/(a^n + b^n).$$

We can teach this to Maple, and try it out when $a = 10$ and $b = 20$ as follows:

```
> x := (n) -> (a^(n+1) + b^(n+1))/(a^n + b^n);
> y := (n) -> (a^n * b + a * b^n)/(a^n + b^n);
> a := 10.0;
> b := 20.0;
> seq(x(n),n=1..50);
> seq(y(n),n=1..50);
```

What do you think are the limiting values? Experiment with other values of a and b if you wish, work out the general rule, and then prove it using the Algebra of Limits.

Wilson's theorem

Investigate the numbers $n!$ (mod p), where p is prime.

(a) Start with a reasonably large prime p, say $p = 19$. Choose a random value of n, say $n = 10$, and then work out $n!$ (mod p) like this:

```
> modp(10!,19);
```

Repeat this for some more values of n.

(b) Now do the same thing more systematically, by listing the sequence of values of $n!$ (mod 19) for $n = 1, \dots, 100$:

```
> seq(modp(n!,19),n=1..100);
```

There is a very obvious pattern when n is large. When exactly does this pattern start? What do you see shortly before the pattern starts?

(c) Repeat part (b) for some other prime numbers in place of 19. You can get a list of the first 20 prime numbers like this:

```
> seq(ithprime(i),i=1..20);
```

Write down your conclusions. Your answer should consist of complete sentences, and should be essentially self-contained, and comprehensible to someone who has not read the question. You should distinguish clearly between things that you can prove and things that you are merely guessing. (It does not matter if you cannot prove anything, provided that you say so.)

(d) Explain the pattern for large n.

You should have noticed something that happens just before the large-n pattern starts. This phenomenon is explained by Wilson's Theorem, which will be proved a little later in the course.

Binomial coefficients

Let $b(n)$ be the greatest common divisor of the numbers $\binom{n}{k}$ for $0 < k < n$. How can we teach Maple this definition?

(1) The binomial coefficient $\binom{n}{k}$ can be expressed as `binomial(n,k)`.
(2) The sequence of binomial coefficients $\binom{n}{k}$ with $0 < k < n$ can be expressed as `seq(binomial(n,k),k=1..n-1)`.
(3) The greatest common divisor of a sequence of integers is given by the function `igcd`.
(4) Thus, the required definition is

```
> b := (n) -> igcd(seq(binomial(n,k),k=1..n-1));
```

We will now investigate the behaviour of this function.

(a) List the values of $b(n)$ for $n = 1, \dots, 100$:

```
> seq(b(n),n=1..100);
```

You should observe than $b(n)$ is usually equal to 1. Our next task is to understand the special values of n where $b(n) \neq 1$.

(b) Roughly in the middle of the list in (a), you should see the number 53. Unfortunately, it is not immediately clear what was the value of n that gave the answer $b(n) = 53$. To cure this, try

```
> seq([n,b(n)],n=1..100);
```

In the middle of the resulting list, we see

$$\dots, [52, 1], [53, 53], [54, 1], \dots.$$

This means that $b(52) = 1$, $b(53) = 53$, and $b(54) = 1$.

(c) Look through the list in (b) for some numbers n where $b(n) \neq 1$. Factorize these numbers n (you can use the `ifactor` function for this). What do you notice?

(d) To be more systematic, we can get Maple to do step (c) for us:

```
> B := (n) -> (b(n) <> 1);
> L := select(B,[seq(n,n=1..100)]);
```

This sets L to be the list of all integers n in the range $n = 1, \dots, 100$ where $b(n) \neq 1$. We can factorize all these integers n as follows:

```
> map(ifactor,L);
```

What do you notice? Your answer should consist of complete sentences, and should be essentially self-contained, and comprehensible to someone who has not read the question. You should distinguish clearly between things that you can prove and things that you are merely guessing. (It does not matter if you cannot prove anything, provided that you say so.)

Fermat's little theorem

In this exercise we investigate the numbers $a^n \pmod{p}$, where p is prime and $0 < a < p$.

(a) Take $p = 7$ and $a = 3$. Calculate 3^n for $n = 0, \dots, 100$:

```
> seq(modp(3^n,7),n=0..100);
```

Examine the resulting sequence carefully; you should notice that it repeats itself. How long is the cycle? What do we get at the beginning of each cycle?

(b) Keeping $p = 7$, repeat part (a) for $a = 2$, then for all other values of a in the range $a = 1, \ldots, 6$.
(c) Now repeat everything for a couple of other primes. What do you notice?

Power sums

For any prime p and any integer $k \geq 0$ we put

$$f(p,k) = 1^k + 2^k + \ldots + (p-1)^k \pmod{p}.$$

We can teach this definition to Maple as follows:

```
> f := (p,k) -> modp(sum(a^k,a=1..p-1),p);
```

Take $p = 7$ and calculate $f(p,k)$ for $k = 0, \ldots, 100$. What do you notice? Try the same thing for some other primes in place of $p = 7$. Write down what you observe, as a complete and self-contained sentence or set of sentences.

Dirichlet's theorem

In this exercise we investigate the numbers $p \pmod{12}$ for all primes p. We will need to use some statistical functions, so we load in Maple's statistical packages:

```
> with(stats);
> with(transform);
```

(a) We can find all prime numbers using the function `ithprime()`; for example, the 20th prime number is `ithprime(20)`, which is 71.
(b) Set P to be the sequence of the first hundred prime numbers, like this:

```
> P := seq(ithprime(i),i=1..100);
```

Now set M to be the sequence of these primes mod 12:

```
> M := seq(modp(ithprime(i),12),i=1..100);
```

Of course, because we are working mod 12, this sequence contains only numbers in the range $0, \ldots, 11$. For each of the numbers in this range, say whether
(i) it occurs repeatedly in M; or
(ii) it occurs only once in M; or
(iii) it does not occur in M at all.
There is a simple rule in terms of prime factors that says which numbers occur repeatedly; can you see it?
(c) Explain why 0 does not occur in the sequence M.
(d) Explain why 4 not occur.
(e) We next look at the numbers that occur repeatedly, and check how many times they occur:

```
> tally(M);
```

The entry `Weight(1,22)` in the resulting list means that 1 occurs 22 times in M. The entry `2` just means that 2 occurs precisely once. You should notice that the entries that occur repeatedly occur roughly the same number of times each. The precise and general version of this statement is called Dirichlet's theorem on primes in arithmetic progression; the proof is hard, but it may be covered in a level four course.

With a little work, we can display this graphically. (Unfortunately the peculiarities of Maple's statistical plotting functions make this harder than it should be.)

```
> ranges := [seq(i-0.5..i+0.5,i=0..11)]:
> T := tallyinto(M,ranges):
> histogram(T,area=1,numbars=12,xtickmarks=12);
```

To get a clearer picture, you can repeat the process with the first thousand (or even ten thousand) primes.

(f) Repeat parts (a), (b) and (e) working modulo 15 instead of modulo 12. Before you start, try to predict what will happen in as much detail as possible.